// File: contracts/flappy.sol



// OPEN SOURCE LICENSE.  RADICAL ACCELERATION.  BREAK THINGS RESPONSIBLY.
// FLAPPY ROYALE:  A TRUSTLESS IMPLEMENTATION FOR EVM BASED BATTLE ROYALE GAMES.
//                                     ................................                                
//                                     @@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@*                               
//                                     @@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@*                               
//                          @@@@@@@@@@@.....................(@@@@@     #@@@@@                          
//                          @@@@@@@@@@@.....................(@@@@@     #@@@@@                          
//                    #@@@@@...........................@@@@@(                @@@@@,                    
//                    #@@@@@...........................@@@@@(                @@@@@,                    
//         /@@@@@@@@@@@@@@@@@@@@@*.....................@@@@@(          #@@@@@     &@@@@@               
//         /@@@@@@@@@@@@@@@@@@@@@*.....................@@@@@(          #@@@@@     &@@@@@               
//    @@@@@#.....................&@@@@@................@@@@@(          #@@@@@     &@@@@@               
//    @@@@@#.....................&@@@@@................@@@@@(          #@@@@@     &@@@@@               
//    @@@@@#.....................%&&&&&,,,,,...........&&&&&(.....     (&&&&&     &@@@@@               
//    @@@@@#...........................@@@@@................(@@@@@                &@@@@@               
//    @@@@@#...........................@@@@@................(@@@@@                &@@@@@               
//    @@@@@#...........................@@@@@......................@@@@@@@@@@@@@@@@@@@@@@@@@@@          
//    @@@@@#...........................@@@@@......................@@@@@@@@@@@@@@@@@@@@@@@@@@@          
//         /@@@@@................&@@@@@.....................(@@@@@***************************@@@@@@    
//         /@@@@@................&@@@@@.....................(@@@@@***************************@@@@@@    
//               @@@@@@@@@@@@@@@@*.....................@@@@@%*****@@@@@@@@@@@@@@@@@@@@@@@@@@@          
//               @@@@@@@@@@@@@@@@*.....................@@@@@%*****@@@@@@@@@@@@@@@@@@@@@@@@@@@          
//                          @@@@@*..........................(@@@@@**********************@@@@@          
//                          @@@@@*..........................(@@@@@**********************@@@@@          
//                          &&&&&/**********................/&&&&&/////////////////////(@@@@@          
//                               &@@@@@@@@@@......................@@@@@@@@@@@@@@@@@@@@@@@@@@@          
//                               &@@@@@@@@@@......................@@@@@@@@@@@@@@@@@@@@@@@@@@@          
//                                          @@@@@@@@@@@@@@@@@@@@@@                                     
//                                          @@@@@@@@@@@@@@@@@@@@@@    
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/Pausable.sol";
import "@openzeppelin/contracts/utils/math/SafeMath.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";

contract FlappyRoyale is Ownable, Pausable, ReentrancyGuard {
    using SafeMath for uint256;

    uint256 public price = 0.02 ether;
    uint256 public highstakePrice = 0.06 ether;
    bool public highStakesActive = false;
    uint256 public entrants = 100;
    uint256 public totalContributions = 0;
    uint256 public totalPrizes = 0;
    uint256 public unclaimedPrizes = 0;
    uint256 public withdrawnFunds = 0;
    uint256 public prizeEquation = 60; 
    uint256 public gameCounter = 0;
    uint256 public currentPlayerCount = 0; 
    address[] public currentPlayersList;  
    address public treasuryWallet = msg.sender;
    uint256 public lastGiveawayMint;

    uint256 public firstPrize = 40;
    uint256 public secondPrize = 20;
    uint256 public thirdPrize = 10;

    mapping(address => uint256) public tickets;
    mapping(address => uint256) public highStakeTickets;
    mapping(address => uint256) public deductedTickets;
    mapping(address => uint256) public pendingWithdrawals; 
    mapping(address => bool) public currentPlayers;  
    mapping(address => uint256) public winsFirst;
    mapping(address => uint256) public winsSecond;
    mapping(address => uint256) public winsThird;

    event LobbyFull(address[] players); 
    event LobbyFullHighStakes(address[] players); 
    event TicketDeducted(address indexed user, uint256 currentPlayerCount);
    event HighStakeTicketDeducted(address indexed user, uint256 currentPlayerCount);
    event TicketsPurchased(address indexed user, uint256 amount);
    event HighStakeTicketsPurchased(address indexed user, uint256 amount);
    event TicketsGranted(address indexed user1, address indexed user2);
    event TicketsGifted(address indexed sender, address indexed receiver, uint256 amount);
    event WinnersPayoutUpdated(address winner1, address winner2, address winner3, uint256 winner1Amount, uint256 winner2Amount, uint256 winner3Amount);
    event EntrantsUpdated(uint256 newEntrants);
    event FundsWithdrawn(uint256 amount);
    event PrizeEquationUpdated(uint256 newPrizeEquation);
    event AmountClaimed(address indexed user, uint256 amount);

    bool public isShutDown = false;
    bool public contractOpened = false;

    modifier onlyWhenHighStakesInactive() {
        require(!highStakesActive, "High Stakes are active");
        _;
    }

    modifier onlyWhenHighStakesActive() {
        require(highStakesActive, "High Stakes are not active");
        _;
    }

    modifier onlyEOA() {
        require(!isContract(msg.sender), "Caller must be an EOA");
        _;
    }

    modifier onlyTreasuryWallet() {
        require(msg.sender == treasuryWallet, "Not authorized");
        _;
    }

    function isContract(address account) public view returns (bool) {
        uint256 size;
        assembly { size := extcodesize(account) }
        return size > 0;
    }
    function toggleHighStakes() external onlyTreasuryWallet {
        require(currentPlayerCount == entrants, "Current players list is not equal to entrants");
        highStakesActive = !highStakesActive;
    }

    function purchaseTickets(uint256 amount) external payable whenNotPaused  {        
        require(!isShutDown, "Ticket purchasing has been shut down");
        require(msg.value == amount.mul(price), "Incorrect ETH sent");

        totalContributions = totalContributions.add(msg.value);
        creditTickets(msg.sender, amount);

        emit TicketsPurchased(msg.sender, amount);
    }

    function creditTickets(address user, uint256 amount) internal {
        tickets[user] = tickets[user].add(amount);
    }

    function purchaseHighStakeTickets(uint256 amount) external payable whenNotPaused  {        
        require(!isShutDown, "Ticket purchasing has been shut down");
        require(msg.value == amount.mul(highstakePrice), "Incorrect ETH sent"); 
        
        totalContributions = totalContributions.add(msg.value);
        creditHighStakeTickets(msg.sender, amount);
        
        emit HighStakeTicketsPurchased(msg.sender, amount);
    }

    function creditHighStakeTickets(address user, uint256 amount) internal {
        highStakeTickets[user] = highStakeTickets[user].add(amount);
    }

    function deductTickets() external onlyEOA onlyWhenHighStakesInactive {        
        require(!isShutDown, "Game is shut down");
        require(contractOpened, "Contract not open yet");
        require(currentPlayerCount < entrants, "Lobby is full. Wait for payout.");
        require(tickets[msg.sender] > 0, "Insufficient tickets");
        require(gameCounter < 1, "Game is in progress please wait");
        require(!currentPlayers[msg.sender], "You've already joined the current lobby");

        if(!currentPlayers[msg.sender]) {
            tickets[msg.sender] = tickets[msg.sender].sub(1);
            deductedTickets[msg.sender] = deductedTickets[msg.sender].add(1);
            
            currentPlayerCount++;
            emit TicketDeducted(msg.sender, currentPlayerCount);
      
            currentPlayersList.push(msg.sender);
            currentPlayers[msg.sender] = true;

            if(currentPlayerCount == entrants) {
                emit LobbyFull(currentPlayersList);
                gameCounter = gameCounter.add(1); 
            }
        }
    }

    function deductHighStakeTickets() external onlyEOA onlyWhenHighStakesActive {        
        require(!isShutDown, "Game is shut down");
        require(contractOpened, "Contract not open yet");
        require(currentPlayerCount < entrants, "Lobby is full. Wait for payout.");
        require(highStakeTickets[msg.sender] > 0, "Insufficient high-stake tickets");
        require(!currentPlayers[msg.sender], "You've already joined the current lobby");

        if(!currentPlayers[msg.sender]) {
            highStakeTickets[msg.sender] = highStakeTickets[msg.sender].sub(1);
            deductedTickets[msg.sender] = deductedTickets[msg.sender].add(1);
        
            currentPlayerCount++;
            emit HighStakeTicketDeducted(msg.sender, currentPlayerCount);
         
            currentPlayersList.push(msg.sender);
            currentPlayers[msg.sender] = true;  
        
            if(currentPlayerCount == entrants) {
                emit LobbyFullHighStakes(currentPlayersList);
                gameCounter = gameCounter.add(1); 
            }
        }
    }

    function giftTickets(address receiver, uint256 amount) external {
        require(receiver != address(0), "Invalid receiver address");
        require(amount > 0, "Amount must be greater than 0");
        address sender = msg.sender;
        require(tickets[sender] >= amount, "Insufficient tickets to gift");

        tickets[sender] = tickets[sender].sub(amount); 
        tickets[receiver] = tickets[receiver].add(amount); 

        emit TicketsGifted(sender, receiver, amount);
    }

    function mintGiftTickets() external onlyTreasuryWallet {
        uint256 amountToMint = 10;
        require(block.timestamp - lastGiveawayMint >= 1 days, 
            "Can only mint 10 tickets per day");

        lastGiveawayMint = block.timestamp;
        tickets[owner()] = tickets[owner()].add(amountToMint);
    }

    function setWinnersPayoutStandard(address winner1, address winner2, address winner3) external onlyTreasuryWallet   {
        _setWinnersPayout(winner1, winner2, winner3, price);
    }

    function setWinnersPayoutHighStakes(address winner1, address winner2, address winner3) external onlyTreasuryWallet   {
        _setWinnersPayout(winner1, winner2, winner3, highstakePrice);
    }

   function _setWinnersPayout(address winner1, address winner2, address winner3, uint256 activePrice) internal {
        require(gameCounter > 0, "No completed games available for payout");
        require(prizeEquation >= 60 && prizeEquation <= 100, "Invalid prize equation percentage");

        uint256 totalPrize = activePrice.mul(entrants); 
        
        uint256 winner1Amount = totalPrize.mul(firstPrize).div(100); 
        uint256 winner2Amount = totalPrize.mul(secondPrize).div(100); 
        uint256 winner3Amount = totalPrize.mul(thirdPrize).div(100);

        pendingWithdrawals[winner1] = pendingWithdrawals[winner1].add(winner1Amount);
        pendingWithdrawals[winner2] = pendingWithdrawals[winner2].add(winner2Amount);
        pendingWithdrawals[winner3] = pendingWithdrawals[winner3].add(winner3Amount);
        totalPrizes = totalPrizes.add(winner1Amount).add(winner2Amount).add(winner3Amount);
        unclaimedPrizes = unclaimedPrizes.add(winner1Amount).add(winner2Amount).add(winner3Amount);

        emit WinnersPayoutUpdated(winner1, winner2, winner3, winner1Amount, winner2Amount, winner3Amount);

        winsFirst[winner1] = winsFirst[winner1].add(1);
        winsSecond[winner2] = winsSecond[winner2].add(1);
        winsThird[winner3] = winsThird[winner3].add(1);

        gameCounter = 0;
        currentPlayerCount = 0;
        delete currentPlayersList;
    }

    function setPrizePercentages(uint256 _firstPrize, uint256 _secondPrize, uint256 _thirdPrize) external onlyTreasuryWallet {
        require(_firstPrize >= 40, "First prize cannot be less than 40%");
        require(_secondPrize >= 20, "Second prize cannot be less than 20%");
        require(_thirdPrize >= 10, "Third prize cannot be less than 10%");
        require(_firstPrize.add(_secondPrize).add(_thirdPrize) <= 100, "Total prize percentages cannot exceed 100%");
        
        firstPrize = _firstPrize;
        secondPrize = _secondPrize;
        thirdPrize = _thirdPrize;
    }


    function claim() external nonReentrant {
        uint256 amount = pendingWithdrawals[msg.sender];
        require(amount > 0, "No funds to claim.");

        pendingWithdrawals[msg.sender] = 0;
        payable(msg.sender).transfer(amount);
        unclaimedPrizes = unclaimedPrizes.sub(amount);

        emit AmountClaimed(msg.sender, amount);
    }

    function updatePrizeEquation(uint256 newPrizeEquation) external onlyTreasuryWallet {
        require(newPrizeEquation >= 60 && newPrizeEquation <= 100, "Invalid prize equation percentage");

        prizeEquation = newPrizeEquation;

        emit PrizeEquationUpdated(newPrizeEquation);
    }

    function grantTickets(address user1, address user2) external onlyTreasuryWallet {
        require(user1 != address(0) && user2 != address(0), "Invalid address");
        tickets[user1] = tickets[user1].add(1);
        tickets[user2] = tickets[user2].add(1);
        emit TicketsGranted(user1, user2);
    }


    function updateEntrants(uint256 newEntrants) external onlyTreasuryWallet {
        if(contractOpened) {
            require(gameCounter > 0, "No completed games available for payout");
        }
        entrants = newEntrants;
        emit EntrantsUpdated(newEntrants);
    }
    
    function getCurrentPlayersList() external view returns (address[] memory) {
        return currentPlayersList;
    }

    function ownerWithdraw(uint256 amount) external onlyOwner {
        uint256 maxAmount = totalContributions.mul(40).div(100);
        require(withdrawnFunds.add(amount) <= maxAmount, "Exceeding withdrawal limit");

        withdrawnFunds = withdrawnFunds.add(amount);
        payable(msg.sender).transfer(amount);

        emit FundsWithdrawn(amount);
    }

    function delegateTreasuryWallet(address newTreasuryWallet) external onlyOwner {
        require(newTreasuryWallet != address(0), "Invalid address");
        treasuryWallet = newTreasuryWallet;
    }

    function renounceOwnership() public view override onlyOwner {
        revert("Renouncing ownership is not allowed");
    }

    function pauseTicketBuying() external onlyTreasuryWallet {
        require(!isShutDown, "Contract is already shut down");
        isShutDown = true;
    }

    function resumeTicketBuying() external onlyTreasuryWallet {
        require(isShutDown, "Contract is not shut down");
        isShutDown = false;
    }

    function startContract() external onlyOwner {
        contractOpened = true;
    }
}
// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/utils/math/SafeMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/math/SafeMath.sol)

pragma solidity ^0.8.0;

// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}